

```
1 ' -----
2 '       From the book Inside Active Directory, ISBN 0-201-61621-1
3 '       Copyright (C) 2002 by Addison-Wesley
4 '       Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim objExcel, objDSE, objSchema, objChild, i, varValue
14 Dim arrParamList(18)
15
16 'Names and Identifiers
17 arrParamList(0) = "LDAPDisplayName"
18 arrParamList(1) = "cn"
19 arrParamList(2) = "adminDisplayName"
20 arrParamList(3) = "distinguishedName"
21 arrParamList(4) = "adminDescription"
22 arrParamList(5) = "attributeID"
23 arrParamList(6) = "mAPIID"
24 arrParamList(7) = "linkID"
25 'Syntax and Content Rules
26 arrParamList(8) = "attributeSyntax"
27 arrParamList(9) = "oMSyntax"
28 arrParamList(10) = "isSingleValued"
29 arrParamList(11) = "rangeLower"
30 arrParamList(12) = "rangeUpper"
31 arrParamList(13) = "extendedCharsAllowed"
32 'Searches
33 arrParamList(14) = "searchFlags"
34 arrParamList(15) = "isMemberofPartialAttributeSet"
35 'Miscellaneous
36 arrParamList(16) = "systemOnly"
37 arrParamList(17) = "systemFlags"
38 arrParamList(18) = "objectCategory"
39
40 Set objExcel = WScript.CreateObject("Excel.Application")
41 objExcel.Visible = True
42 objExcel.Workbooks.Add
43 objExcel.ActiveSheet.Name = "attributeSchema objects"
44
45 objExcel.ActiveSheet.Range("A1").Activate
46 For i = LBound(arrParamList) To UBound(arrParamList)
47     objExcel.ActiveCell.Offset(0,i).Value = arrParamList(i)
48 Next
49 objExcel.ActiveCell.Offset(1,0).Activate    'Move to next line
50
51 Set objDSE = GetObject("LDAP://rootDSE")
52 Set objSchema = GetObject("LDAP://" & _
53     objDSE.Get("schemaNamingContext"))
54
55 On Error Resume Next
56 For Each objChild In objSchema
57     If objChild.Class = "attributeSchema" Then
58         For i = LBound(arrParamList) To UBound(arrParamList)
59             varValue = objChild.Get(arrParamList(i))
60             If Err <> 0 Then    'property not found or some other error
61                 Err.Clear
62                 varValue = ""
63             End If
64             objExcel.ActiveCell.Offset(0,i).Value = varValue
65         Next
66         objExcel.ActiveCell.Offset(1,0).Activate    'Move to next line
67     End If
68 Next
69
```